

R. Brigola, TH Nürnberg Georg Simon Ohm, 2015

Mathematica - Notebooks als Bonusmaterial zum Lehrbuch

[1] Rolf Brigola Fourier-Analysis und Distributionen,
Eine Einführung mit Anwendungen,
edition swk, Hamburg 2013

Demo-Material über lineare Differentialgleichungen mit konstanten
Koeffizienten, Anwendungsbeispiele des Distributionenkalküls

Referenzen zu Kapiteln, Abschnitten, Seitenzahlen beziehen sich im Folgenden meist auf das genannte Lehrbuch. Einige wenige weitere Referenzen findet man am Ende des Notebooks.

Die URL aller meiner *Mathematica*-Notebooks zur Fourier-Analysis ist:
<http://www.stiftung-swk.de/mathematica/>

Ziel diese Notebooks ist, Nutzern wie Teilnehmerinnen in meinen Lehrveranstaltungen einige Beispiele zu *Anwendungen von Distributionen bei linearen Differentialgleichungen mit konstanten Koeffizienten* mit Hilfe von *Mathematica* vorzustellen. Ich gehe dabei davon aus, dass Nutzern mein vorangehendes Notebook über einigen Grundlagen zur Distributionentheorie schon bekannt ist. Man findet es (und weitere Demo-Notebooks) unter folgender URL des Autors: www.stiftung-swk.de/mathematica

Das Notebook ist mit *Mathematica* 10 unter Windows 7 erstellt.

In[189]=

```
ClearAll["Global`*"]; Remove["Global`*"]; ?Global`*  
SetDirectory["D:/Mathematica"] (*User-spez. Verzeichnis*)
```

Information::nomatch : No symbol matching Global`* found. >>

Out[190]= D:\Mathematica

1. Grundlösungen linearer Differentialgleichungen mit konstanten Koeffizienten

Für einen gegebenen gewöhnlichen Differentialoperator $P(D)$ nennt man eine Lösung g eine **Grundlösung** oder auch **Fundamentallösung**, wenn $P(D)u = \delta$ gilt. Dabei ist δ die Dirac-Distribution und P ein Polynom. Für $P(z) = \sum_{k=0}^n a_k z^k$ ist also $P(D)u = \sum_{k=0}^n a_k \partial^k u$. P ist das *charakteristische Polynom* der DGL. Alle Ableitungen sind als verallgemeinerte Ableitungen aufzufassen, d.h. die DGL ist eine Gleichung zwischen Distributionen. Es gibt i.A. unendlich viele Grundlösungen, aber genau eine kausale. "**Kausal**" bedeutet, dass diese Lösung ihren Träger in $[0, \infty[$ hat. Versteht man die DGL $P(D)u = \delta$ als Beschreibung

eines Übertragungssystems, das einen δ -Impuls als Input erhält, dann bedeutet kausal also, dass das System erst nach dem Input einen Output erzeugt. Für den Fall, dass auf der rechten Seite der Gleichung kein Differentialoperator steht, nennen wir die kausale Grundlösung auch kurz **“die Impulsantwort”** des Übertragungssystems.

Man findet die kausale Grundlösung durch Lösung des Anfangswertproblems für die homogene DGL $P(D)u=0$ zu den Anfangswerten $u(0) = u'(0) = \dots = u^{(n-2)}(0)=0$ und $u^{(n-1)}(0) = 1/a_n$.

Ist u diese Lösung, dann ist $u(t)$ Heaviside(t) die gesuchte kausale Grundlösung.

Für einen Beweis dieses Satzes sei auf [1], Kap. 8 verwiesen. Insbesondere treten in u also Lösungsanteile der homogenen DGL auf.

Kennt man die kausale Grundlösung g , dann findet man die **kausale Lösung für $P(D)u=f$** für kausale f (d.h. f mit $\text{Tr}(f)$ in $[0, \infty[$) **durch die Faltung $g * f$** von g mit f .

Dieses Grundlösungsverfahren bildet damit die Basis für die Lineare Systemtheorie zeitinvarianter Systeme, hier durch Rechnen im “Zeitbereich”, wodurch mit Hilfe der Fouriertransformation in der Folge die für Anwendungen so wichtigen abgeleiteten Begriffe “im Frequenzbereich” erarbeitet werden können.

1.1 Beispiele für das Grundlösungsverfahren bei linearen DGL'en mit konst. Koeffizienten Faltung mit der kausalen Grundlösung bei inhomogenen Problemen

Beispiel 1. Wir berechnen die kausale Grundlösung für $y''+y=\delta$ mit Mathematica

```
In[191]= DSolve[{y''[t] + y[t] == 0, y[0] == 0, y'[0] == 1}, y[t], t]
          grdlsol[t_] = %[[1, 1, 2]] HeavisideTheta[t]
```

```
Out[191]= {{y[t] -> Sin[t]}}
```

```
Out[192]= HeavisideTheta[t] Sin[t]
```

Das zur DGL gehörige Übertragungssystem ist instabil, wie man an der Lage der Nullstellen des charakteristischen Polynoms sofort erkennt.

Hier die allgemeine Lösung für $y''+y=\delta$ direkt mit Mathematica distributionell gerechnet:

```
In[193]= DSolve[y''[t] + y[t] == DiracDelta[t], y[t], t]
```

```
Out[193]= {{y[t] -> C[1] Cos[t] + C[2] Sin[t] + HeavisideTheta[t] Sin[t]}}
```

Beispiel 2. Wir stellen die DGL für ein **Butterworth-Tiefpassfilter 3. Ordnung** auf, berechnen die zugehörige *Impulsantwort* g und die daraus resultierende *Sprungantwort* $g * \text{UnitStep}$.

Zu Butterworth-Tiefpassfiltern vgl. man das Notebook auf dieser URL, das schon zur Verfügung steht. Dort findet man Einzelheiten zum Entwurf von Butterworth-Filtern und Hilfsmittel, die in Mathematica bereits implementiert sind.

Das charakteristische Polynom der DGL eines Butterworthfilters 3. Ordnung ist

$$P(z) = (1 + z/\omega_g)(1 + z/\omega_g + z^2/\omega_g^2)$$

mit der Grenzkreisfrequenz ω_g . $1/P(z)$ nennt man die **Übertragungsfunktion des Filters**, $1/P(i\omega)$ ist bei stabilen Systemen wie hier dann der **Frequenzgang des Filters**, aufzufassen als **komplexwertige Distribution aus S'** wie im vorangehenden Notebook mit Hilfe des Cauchy-Hauptwerts schon erklärt wurde.

Wir lösen mit *Mathematica* direkt die distributionelle DGL $P(D)u=\delta$ und setzen dabei wegen der Kausalitätsbedingung die Konstanten in der allgemeinen Lösung Null. Um konkret zu werden verwenden wir 4.2 kHz als Grenzkreisfrequenz des Filters (vgl. auch [1], Beispiel 4 auf S. 279).

In[194]=

```
 $\omega_g := 8400 \pi$  (* Grenzkreisfrequenz *)
g[t_] =
Simplify[DSolve[y''''[t]/ $\omega_g^3 + 2 y''[t]/\omega_g^2 + 2 y'[t]/\omega_g + y[t] = \text{DiracDelta}[t]$ ,
y[t], t][[1, 1, 2]]] /. {C[1] -> 0, C[2] -> 0, C[3] -> 0} (* Impulsantwort *)
```

Out[195]= $-2800 e^{-8400 \pi t} \pi \text{HeavisideTheta}[t]$
 $\left(-3 + 3 e^{4200 \pi t} \text{Cos}[4200 \sqrt{3} \pi t] - \sqrt{3} e^{4200 \pi t} \text{Sin}[4200 \sqrt{3} \pi t]\right)$

Die nachfolgende Berechnung der Sprungantwort mit dem Befehl **Convolve** ist etwas zeitaufwändig, (warum auch immer, da diese Faltung eigentlich auch per Hand schnell auszurechnen wäre).

In[196]= `stepresponse[t_] = Convolve[g[s], UnitStep[s], s, t]` (* Sprungantwort *)

Out[196]= $\frac{1}{3} \text{HeavisideTheta}[t] \left(3 - 3 e^{-8400 \pi t} - 2 \sqrt{3} e^{-4200 \pi t} \text{Sin}[4200 \sqrt{3} \pi t]\right)$


Die Fouriertransformierte der Impulsantwort ist der Frequenzgang dieses Butterworthfilters. Im Nenner ist wieder das charakteristische Polynom der DGL zu erkennen. Man muss dort nur $i\omega$ durch z ersetzen, durch den Zähler dividieren und erhält $P(z)$.

In[197]= `frgang[$\omega_$] = FourierTransform[g[t], t, ω , FourierParameters -> {1, -1}]`


Out[197]= $(592704000000 \pi^3) / ((8400 \pi + i \omega) (70560000 \pi^2 + 8400 i \pi \omega - \omega^2))$

Nachfolgend nun die Plots der Impuls- und der Sprungantwort (die Zeitachse in ms), des Amplitudengangs und des Phasengangs des Filters (die Grenzkreisfrequenz ist auf Eins normiert):


In[198]= `Plot[g[t/1000], {t, 0, 1}, PlotStyle -> {Blue, Thickness[0.006]},
PlotRange -> {-2000, 11000}, Frame -> True,
FrameLabel -> "t in ms"]` (* Impulsantwort *)




```
In[199]:= Plot[stepresponse[t/1000], {t, 0, 2},  
  PlotStyle -> {Blue, Thickness[0.006]}, PlotRange -> {-0.1, 1.2},  
  Frame -> True, FrameLabel -> "t in ms"] (* Sprungantwort *)
```



```
In[200]:= Plot[Abs[fgang[s ωg]], {s, 0, 4}, PlotStyle -> {Blue, Thickness[0.006]},  
  PlotRange -> {-0.1, 1.1}, Frame -> True, FrameLabel -> "s ωg, s ∈ [0,4]"  
  (* Amplitudengang, Grenzkreisfrequenz auf 1 normiert *)
```



```
In[201]:= Plot[Arg[frgang[s ωg]], {s, 0, 4},
  PlotStyle → {Blue, Thickness[0.006]}, PlotRange → {-π, π},
  Frame → True, FrameLabel → "s ωg, s ∈ [0,4]" (* Phasengang *)
```



Anmerkung zum Thema “Mathematische Modelle versus physikalische Realität” oder “Warum Distributionentheorie so nützlich ist”:

Die Beschreibung physikalischer Zusammenhänge wie oben eines Übertragungssystems, das in der Realität durch eine elektrotechnische Schaltung ausgeführt wird, durch *lineare DGL'en* und durch Lösungen für *rechte Seiten wie einen Dirac-Impuls oder eine Sprungfunktion*, ist ein *mathematisches Modell*, d.h. ein Konstrukt des menschlichen Geistes mit idealen Objekten und Begriffen wie den Begriff der Ableitung (ein Limes von Differenzenquotienten) oder die Begriffe und den Kalkül der Distributionen. Sobald man damit arbeitet, *arbeitet man als Mathematiker mit Idealisierungen, die es in der dinglichen Realität nicht gibt*.

Niemand kann einen Dirac-Impuls in der Realität erzeugen, denn er müsste unendliche Energie besitzen, niemand kann je einen Gleichspannungsverlauf mit einer Sprungstelle wie beim UnitStep erzeugen (das bräuchte unendliche Bandbreite), immer werden in realen Systemen stetige Übergänge vorhanden sein. Selbst die *linearen DGL'en* zur Beschreibung gewisser Sachverhalte sind schon Idealisierungen, keine reale Schaltung wird je vollkommen linear arbeiten und keine Koeffizienten der DGL werden vollkommen konstant sein.

Bei der Umsetzung der Idealisierungen - z.B. in eine Filterschaltung - muss nach der berechnenden Arbeit als Mathematiker ein *Rollenwechsel* hin zur Arbeit z.B. als Physiker oder Ingenieur stattfinden, der *ein berechnetes ideales System (unseres Geistes) näherungsweise in der Realität nachzubilden* versucht.

Dieser Rollenwechsel macht Studierenden in den Anfängen erfahrungsgemäß Schwierigkeiten, wenn nicht immer klar ist, ob sie gerade als Mathematiker oder als Praktiker arbeiten und mag (leider) dazu beitragen, dass sog. Praktiker sich nur mit Vorbehalten originär mathematischer Arbeit zuwenden, um auch auf diesem Gebiet Neues, d.h. neue Theorie zu erlernen. Ohne mathematische Theorie aber gäbe es auch neue technische Entwicklungen wohl sicher nicht.

Um beim Thema dieses Notebooks zu bleiben, sei ein Beispiel gegeben: Statt mit komplizierten Beschreibungen realer Verhältnisse an einem Schwingkreis zu rechnen, etwa Spannungsverläufen,


die zwar einen Impuls oder eine plötzliche Aufschaltung von Gleichspannung beschreiben, aber dennoch vollkommen glatt (beliebig differenzierbar) sind (man vgl. das nachfolgende Beispiel mit der ganzen Rechenpower von *Mathematica*), stellt die *Distributionentheorie einfache Modelle wie einen Dirac-Impuls oder eine Sprungfunktion und eine einfache Lösungstheorie für lineare DGL'en mit konstanten Koeffizienten zur Verfügung*. Der Preis, den man dafür zu zahlen hat, ist nur das Erlernen der Theorie- und Kalkülgrundlagen. In der Praxis versucht man dann, die Modelle so gut wie möglich nachzubilden. Hier wird weder eine ideale lineare DGL gelten, ihre Koeffizienten (Widerstände, Kapazitäten etc.) werden nicht wirklich konstant sein, es gibt keinen Dirac-Impuls und auch keine Sprungfunktion. Würde man aber Theorie vermeiden wollen, müsste man den wohl viel höheren Preis bezahlen, dass dann viel komplizierter zu rechnen wäre und auch einfache beschreibende Begriffe nicht so zur Verfügung ständen, wie wir es heute kennen.

Beispiel 3. Nachfolgend zur Illustration der Versuch, unsere letzte DGL - immer noch eine Idealisierung - mit einer "Impulsnachbildung" durch eine glatte Funktion zu lösen. Ich verwende dazu einmal die Funktion $f(t) = n / (\pi(1 + (nt)^2))$, die häufig in Einführungen des Dirac-Impulses als δ -Folge verwendet wird, und versuche, mit *Mathematica* die Gleichung $P(D)u=f$ für $n=40$ zu lösen, $P(D)$ wie in Beispiel 2.

An der Lösung, die *Mathematica* zwar bietet, wird man wohl schwerlich Eigenschaften unseres Butterworthfilters ablesen oder daraus durch Fouriertransformation den Frequenzgang des Filters bilden können. Sehr viel einfacher also, etwas Theorie zu lernen und als Gewinn dafür einfach rechnen zu können. Um mit dem berühmten Physiker G. Kirchhoff zu sprechen:

"Es gibt nichts Praktischeres als eine gute Theorie!"

```
In[202]:=  $\omega_g := 8400 \pi$  ; (* Grenzkreisfrequenz *)
f[t_] := 40 / ( $\pi (1 + 1600 t^2)$ );
Plot[f[t], {t, -1, 1}, PlotRange -> {{-1, 1}, {-4, 15}}, Frame -> True,
FrameStyle -> Directive[Black, FontSize -> 22, FontWeight -> Plain],
PlotStyle -> {Blue, Thickness[0.006]}]
```



```
In[205]:= gtest[t_] =
  Simplify[DSolve[y''''[t]/\omega_g^3 + 2 y'''[t]/\omega_g^2 + 2 y''[t]/\omega_g + y[t] == f[t], y[t], t][[
    1, 1, 2]]] /. {C[1] -> 0, C[2] -> 0, C[3] -> 0}
  (* das wäre die Lösung als Antwort auf obige Impulsnachbildung *)
```

```
Out[205]= e^{-105 \pi (\sqrt{3} + 80 t)} \left( -4200 i e^{105 \sqrt{3} \pi} \text{ExpIntegralEi} [210 \pi (-i + 40 t)] -
  2100 i e^{4200 \pi t} \text{Cos} [4200 \sqrt{3} \pi t] \text{ExpIntegralEi} [105 (\pi + i \sqrt{3} \pi) (-i + 40 t)] -
  700 \sqrt{3} e^{4200 \pi t} \text{Cos} [4200 \sqrt{3} \pi t] \text{ExpIntegralEi} [105 (\pi + i \sqrt{3} \pi) (-i + 40 t)] +
  4200 i e^{105 \sqrt{3} \pi} \text{ExpIntegralEi} [210 \pi (i + 40 t)] +
  2100 i e^{4200 \pi t} \text{Cos} [4200 \sqrt{3} \pi t] \text{ExpIntegralEi} [105 (\pi - i \sqrt{3} \pi) (i + 40 t)] -
  700 \sqrt{3} e^{4200 \pi t} \text{Cos} [4200 \sqrt{3} \pi t] \text{ExpIntegralEi} [105 (\pi - i \sqrt{3} \pi) (i + 40 t)] +
  2100 i e^{210 \pi (\sqrt{3} + 20 t)} \text{Cos} [4200 \sqrt{3} \pi t] \text{ExpIntegralEi} [105 (\pi + i \sqrt{3} \pi) (i + 40 t)] +
  700 \sqrt{3} e^{210 \pi (\sqrt{3} + 20 t)} \text{Cos} [4200 \sqrt{3} \pi t] \text{ExpIntegralEi} [105 (\pi + i \sqrt{3} \pi) (i + 40 t)] +
  700 (-3 i + \sqrt{3}) e^{210 \pi (\sqrt{3} + 20 t)} \text{ExpIntegralEi} [105 (\pi - i \sqrt{3} \pi) (-i + 40 t)]
  \left( \text{Cos} [4200 \sqrt{3} \pi t] + i \text{Sin} [4200 \sqrt{3} \pi t] \right) -
  2100 e^{4200 \pi t} \text{ExpIntegralEi} [105 (\pi + i \sqrt{3} \pi) (-i + 40 t)] \text{Sin} [4200 \sqrt{3} \pi t] +
  700 i \sqrt{3} e^{4200 \pi t} \text{ExpIntegralEi} [105 (\pi + i \sqrt{3} \pi) (-i + 40 t)] \text{Sin} [4200 \sqrt{3} \pi t] -
  2100 e^{4200 \pi t} \text{ExpIntegralEi} [105 (\pi - i \sqrt{3} \pi) (i + 40 t)] \text{Sin} [4200 \sqrt{3} \pi t] -
  700 i \sqrt{3} e^{4200 \pi t} \text{ExpIntegralEi} [105 (\pi - i \sqrt{3} \pi) (i + 40 t)] \text{Sin} [4200 \sqrt{3} \pi t] +
  2100 e^{210 \pi (\sqrt{3} + 20 t)} \text{ExpIntegralEi} [105 (\pi + i \sqrt{3} \pi) (i + 40 t)] \text{Sin} [4200 \sqrt{3} \pi t] -
  700 i \sqrt{3} e^{210 \pi (\sqrt{3} + 20 t)} \text{ExpIntegralEi} [105 (\pi + i \sqrt{3} \pi) (i + 40 t)] \text{Sin} [4200 \sqrt{3} \pi t] \right)
```

1.2 Anfangswertprobleme bei Differentialgleichungen $P(D)u=Q(D)f$ mit distributionellen rechten Seiten

Häufig findet man in Anwendungen der Linearen Systemtheorie Gleichungen der Form

$$P(D)u=Q(D)f,$$

d.h. auch auf der rechten Seite der DGL steht ein Differentialoperator $Q(D)$. Die Funktion oder Distribution f ist dann der System-Input, die Lösung u (bei kausalen Systemen gibt es nur eine) der System-Output. Außerdem spielen in Anwendungen auch Anfangswertprobleme eine Rolle. Bei Distributionen in den Gleichungen bedarf es allerdings zusätzlicher Voraussetzungen, damit überhaupt sinnvoll von Anfangswerten gesprochen werden kann, da etwa ein Dirac-Impuls ja keinen Wert annimmt. Dies sind i.d.R. Differenzierbarkeitsbedingungen an die Funktion f auf der rechten Seite der Gleichung. Zur Definition, wie ein *kausales Anfangswertproblem für $P(D)u=Q(D)f$* verstanden werden kann und wie seine Lösung aussieht, verweise ich auf [1], Abschnitt 8.2.

Wir behandeln hier mit *Mathematica 2* Beispiele solcher gewöhnlichen linearen DGL'en mit konstanten Koeffizienten.

Beispiel 1. Für die geradlinige Bewegung $x(t)$ eines zunächst kräftefreien Teilchens der Masse m gelte $x(0^-) = x_0$, $x'(0^-) = v_0$. Zur Zeit $t=0$ wirke eine Stoßkraft, die mathematisch durch die Dirac-Distribution $Z\delta$ modelliert sei. Die Konstante Z gibt die Impulsänderung in Ns an. Das Modell erklärt hier an einem Beispiel der Mechanik, warum δ auch als Impuls bezeichnet wird. Die Bewegungsgleichung lautet

$$mx'' = Z \delta, \quad x(0^-) = x_0, \quad x'(0^-) = v_0$$

Auf ganz \mathbb{R} hat das kausale Anfangswertproblem die eindeutige Lösung

$$x(t) = \frac{Z}{m} t \text{ HeavisideTheta}(t) + v_0 t + x_0.$$

Der erste Anteil der linken Seite beschreibt die Wirkung der Stoßkraft ab $t=0$, der andere Anteil - die Lösung des zugehörigen homogenen AWP - beschreibt die Nachwirkung der Vergangenheit, die zu den Anfangswerten geführt hat.

Oft (und sinnvollerweise) interessiert man sich nur für die Lösung ab $t=0$ unter "**Ausblendung der Vergangenheit**". Die Lösung dieses AWP als Distribution in D_+' , der Faltungsalgebra aller Distributionen mit Träger in $[0, \infty[$, ist dann

$$T(t) = g * Z\delta(t) + (v_0 t + x_0) \text{ Heaviside}(t) \text{ mit der Grundlösung } g \text{ von } P(D)x = m x'' = \delta.$$

Wir rechnen mit konkreten Werten für m , Z , x_0 und v_0 .

Wir sehen eine abrupte Geschwindigkeitsänderung ab $t=0$ als Folge der Stoßkraft. Die Lösung für $t \rightarrow -\infty$ geht gegen $-\infty$ und wird physikalisch für das Stoßproblem kaum von Belang sein. Daher ist es plausibel, sich nur für die Lösung ab dem Stoß, d.h. für $t \geq 0$ zu interessieren und außer Betracht zu lassen, auf welche Weise die Anfangswerte in der Vergangenheit zustande gekommen sein könnten. Dem entspricht dann, das Problem nur auf der Halbachse $[0, \infty[$ anzusehen.

```
In[206]:= m := 2; Z := 5; x0 := 3; v0 := 0.4;
xhom[t] =
  Simplify[DSolve[{m x''[t] == 0, x[0] == x0, x'[0] == v0}, x[t], t][[1, 1, 2]]]
  (* Lösung des homogenen AWP *)


Out[207]= 3. + 0.4 t

In[208]:= g1[t_] = Simplify[DSolve[m x''[t] == DiracDelta[t], x[t], t][[1, 1, 2]]] /.
  {C[1] -> 0, C[2] -> 0} (* kausale Grundlösung P(D)g1=delta *)

Out[208]= 1/2 t HeavisideTheta[t]
```


```
In[209]:= sol1[t_] = Convolve[g1[s], Z DiracDelta[s], s, t] + xhom[t]
(* Lösung auf R für das kausale AWP *)
Plot[sol1[t], {t, -1., 1.}, PlotStyle -> {Blue, Thickness[0.006]}]
```

```
Out[209]= 3. + 0.4 t +  $\frac{5}{2}$  t HeavisideTheta[t]
```



```
In[211]:= sol2[t_] = Convolve[g1[s], Z DiracDelta[s], s, t] + xhom[t] HeavisideTheta[t]
(* Lösung aus D'+ unter Ausblendung der Vergangenheit nur ab t=0 *)
Plot[sol2[t], {t, -1., 1.}, PlotStyle -> {Blue, Thickness[0.006]}]
```

```
Out[211]= (3. + 0.4 t) HeavisideTheta[t] +  $\frac{5}{2}$  t HeavisideTheta[t]
```



Lösung des Problems auf der Halbachse $[0, \infty[$ mit der entsprechend modifizierten DGL für Distributionen in der Faltungsalgebra $D_+' = \{T \in D' : \text{Tr}(T) \subset [0, \infty[\}$

Man erhält die gleiche Lösung sol2 aus D_+' unmittelbar als Lösung der distributionellen Gleichung


$$m x'' = Z \delta + m (x_0 \delta' + v_0 \delta).$$

Hier sind die Anfangswerte in passender Weise schon "in die Gleichung eingearbeitet". Für mehr Details vgl. man bitte den Satz auf S. 183 in [1], Abschnitt 8.2 oder [2], Abschnitt 9.3.

Auch dies noch mit *Mathematica*, indem wir *Mathematica* direkt die richtig modifizierte Distributionengleichung lösen lassen:

```
In[213]= sol3[t_] = Simplify[DSolve[
    m x''[t] == Z DiracDelta[t] + m (x0 D[DiracDelta[t], t] + v0 DiracDelta[t]),
    x[t], t][[1, 1, 2]]] /. {C[1] -> 0, C[2] -> 0}
Plot[sol3[t], {t, -1., 1.}, PlotStyle -> {Blue, Thickness[0.006]}]
(* Die Lösung in D+' mit Träger in [0,∞[ *)
```

Out[213]= 0. + (3. + 2.9 t) HeavisideTheta[t]



Das gleiche Ergebnis erhält man natürlich auch durch Faltung der kausalen Grundlösung mit der ebenfalls zu D_+ gehörenden rechten Seite $Z \delta + m (x_0 \delta' + v_0 \delta)$ der Gleichung.

```
In[215]= Simplify[Convolve[g1[s],
    Z DiracDelta[s] + m (x0 D[DiracDelta[s], s] + v0 DiracDelta[s]), s, t]]
```


Out[215]= (3. + 2.9 t) HeavisideTheta[t]

Beispiel 2. Wir suchen die Lösung der Differentialgleichung

$$u'' + \frac{2}{\sqrt{LC}} u' + \frac{1}{LC} u = U_1 \delta' \text{ und } u(0^-) = U_0, u'(0^-) = 0.$$

Die Gleichung beschreibt einen RCL-Schwingkreis bei kritischer Dämpfung ($R^2=4L/C$) mit Eingangsspannung U_1 HeavisideTheta(t) und gegebenen Anfangswerten. Die Lösung ist der Spannungsverlauf an der Induktivität (man vgl. [1], Beispiel 3 auf S. 189).

```
In[216]= Import["rcl_circuit.png", ImageSize -> Small]
```



Wir interessieren uns nur für die Lösung ab $t=0$, d.h. wir modifizieren die Gleichung zu einer Gleichung in D_+ zu

$$u'' + \frac{2}{\sqrt{LC}} u' + \frac{1}{LC} u = u_1 \delta' + \frac{2U_0}{\sqrt{LC}} \delta + u_0 \delta'$$

und lösen sie mit *Mathematica*, indem wir die kausale Grundlösung bestimmen und diese mit der rechten Seite falten. (Eine Lösung für $t \rightarrow -\infty$ würde unbegrenzt wachsen und ist daher physikalisch nicht von Bedeutung, die reale Vergangenheit des Schwingkreises reicht ja auch nicht bis ins Mittelalter zurück.)

```
In[217]:= grundloesung[t_] =
  Simplify[DSolve[u''[t] + 2/(Sqrt[LC]) u'[t] + 1/(LC) u[t] == DiracDelta[t],
    u[t], t][[1, 1, 2]] /. {C[1] -> 0, C[2] -> 0}
```

```
Out[217]= e^{-t/sqrt[LC]} t HeavisideTheta[t]
```

```
In[218]:= solution[t_] = Simplify[Convolve[grundloesung[s],
  (U_0 + U_1) D[DiracDelta[s], s] + 2 U_0 / Sqrt[LC] DiracDelta[s], s, t]]
```

```
Out[218]= 1/sqrt[CL] e^{-t/sqrt[CL]} HeavisideTheta[t] ((sqrt[CL] + t) U_0 + (sqrt[CL] - t) U_1)
```

2. Lösung für lineare Systeme 1. Ordnung mit konstanten Koeffizienten

Selbstverständlich kann man Distributionsmethoden auch zur Lösung für lineare Systeme erster Ordnung mit konstanten Koeffizienten verwenden und auch hier wieder AWP'e auf der Halbgeraden $t \geq 0$ durch eine entsprechende Modifikation der Gleichung wie zuletzt lösen. Für ein System 1. Ordnung mit konstanten Koeffizienten der Form

$$X' = A X + F$$

übernimmt die Matrix $G[t] = e^{At} \text{HeavisideTheta}[t]$ die Rolle der kausalen Grundlösung. Die kausale Lösung für eine Inhomogenität mit einer Vektordistribution F mit Träger in $[0, \infty[$ erhält man dann durch die Faltung $G * F$, d.h. den Vektor mit den Komponenten c_k , definiert für $n \times n$ -Matrizen und Vektordistributionen X und F mit n Komponenten durch

$$c_k = \sum_{m=1}^n g_{km} * f_m \quad (k=1, \dots, n).$$

Entsprechend zu den obigen Ausführungen erhält man für ein kausales Anfangswertproblem mit einem Anfangswertvektor x_0 aus \mathbb{R}^n und F aus $(D_+)^n$ die eindeutige Lösung von

$$X' = A X + F + x_0 \delta$$

in $(D_+)^n$ durch Faltung von G mit der rechten Seite dieser Gleichung. Für stetige F ist die Lösung regulär und stimmt für $t > 0$ mit der klassischen Lösung überein, die man durch Variation der Konstanten erhält. Der klassischen Formel dafür ist die verwendete Faltung direkt anzusehen.

Beispiel. Ich beschränke mich auf ein Beispiel, das noch einmal das Beispiel 2 aus 1.1 aufgreift, indem es die dortige DGL für das Butterworthfilter in ein System 1. Ordnung umwandelt ("wie" lernt man in Analysis-Vorlesungen oder Sie finden es in [1]) und dieses wie gerade gesagt löst. Alle Anfangswerte sind dabei Null. In der Ingenieurliteratur nennt man diese Beschreibung "**Zustandsraum-Beschreibung**" des Filters (engl. **state space model**). Das System lautet in der jetzt verwendeten Form, zunächst mit allgemeiner Grenzkreisfrequenz Ω

$X' = A X + F$ mit A und F wie folgt:

```
In[219]:= A := {{0, 0, -Ω³}, {1, 0, -2 Ω²}, {0, 1, -2 Ω}} (* 3x3-Systemmatrix A *)
A // MatrixForm
F := {Ω³ DiracDelta[t], 0, 0} (* Inhomogenität/ Input F *)
F // MatrixForm
```

Out[220]/MatrixForm=

$$\begin{pmatrix} 0 & 0 & -\Omega^3 \\ 1 & 0 & -2 \Omega^2 \\ 0 & 1 & -2 \Omega \end{pmatrix}$$

Out[222]/MatrixForm=

$$\begin{pmatrix} \Omega^3 \text{DiracDelta}[t] \\ 0 \\ 0 \end{pmatrix}$$

```
In[223]:= G := Simplify[ComplexExpand[MatrixExp[A t]]] HeavisideTheta[t]
G // MatrixForm
```

Out[224]/MatrixForm=

$$\begin{pmatrix} \text{HeavisideTheta}[t] \left(e^{-t \Omega} + \frac{2 e^{-\frac{t \Omega}{2}} \sin\left[\frac{1}{2} \sqrt{3} t \Omega\right]}{\sqrt{3}} \right) & -\frac{1}{3} e^{-t \Omega} \Omega \text{HeavisideTheta}[t] \left(3 - 3 e^{\frac{t \Omega}{2}} \cos\left[\frac{1}{2} \sqrt{3} t \Omega\right] \right) \\ \frac{e^{-t \Omega} \text{HeavisideTheta}[t] \left(1 - e^{\frac{t \Omega}{2}} \cos\left[\frac{1}{2} \sqrt{3} t \Omega\right] + \sqrt{3} e^{\frac{t \Omega}{2}} \sin\left[\frac{1}{2} \sqrt{3} t \Omega\right] \right)}{\Omega} & e^{-t \Omega} \left(-1 + 2 e^{\frac{t \Omega}{2}} \cos\left[\frac{1}{2} \sqrt{3} t \Omega\right] \right) \\ \frac{e^{-t \Omega} \text{HeavisideTheta}[t] \left(3 - 3 e^{\frac{t \Omega}{2}} \cos\left[\frac{1}{2} \sqrt{3} t \Omega\right] + \sqrt{3} e^{\frac{t \Omega}{2}} \sin\left[\frac{1}{2} \sqrt{3} t \Omega\right] \right)}{3 \Omega^2} & \frac{e^{-t \Omega} \text{HeavisideTheta}[t] \left(-3 + 3 e^{\frac{t \Omega}{2}} \cos\left[\frac{1}{2} \sqrt{3} t \Omega\right] \right)}{\Omega} \end{pmatrix}$$

Die Komponente g_{31} ergibt die Impulsantwort des Butterworthfilters, die wir oben schon einmal berechnet hatten. Bei der Darstellung von G ist der *Mathematica*-Befehl **ComplexExpand** nützlich, um das Ergebnis übersichtlicher darzustellen (Testen Sie selbst mal ohne diesen Befehl).

```
In[225]:= grdlsgneu[t_] = G[[3, 1]]
```

Out[225]=

$$\frac{1}{3 \Omega^2} e^{-t \Omega} \text{HeavisideTheta}[t] \left(3 - 3 e^{\frac{t \Omega}{2}} \cos\left[\frac{1}{2} \sqrt{3} t \Omega\right] + \sqrt{3} e^{\frac{t \Omega}{2}} \sin\left[\frac{1}{2} \sqrt{3} t \Omega\right] \right)$$

Wir überprüfen die Behauptung, indem wir $\Omega = \omega_g = 8400 \pi$ wie oben setzen und aus der Impulsantwort (durch die Faltung $\text{grdlsgneu} * \Omega^3 \delta$ sofort bestimmt) **durch Fouriertransformation** den hier berechneten Frequenzgang **frgangneu** mit dem oben berechneten **frgang** vergleichen: Wir haben das gleiche Ergebnis. Auf den Zusammenhang zwischen Impulsantwort und Frequenzgang gehe ich in einem anderen Notebook über Anwendungen der Fouriertransformation bei linearen Übertragungssystemen näher ein.

In[226]:= $\Omega := 8400 \pi$

In[227]:= `frgangneu[ω] =`

`FourierTransform[Ω^3 grdlsngneu[t], t, ω , FourierParameters \rightarrow {1, -1}]`

Out[227]= $(592\,704\,000\,000\,\pi^3) / ((8400\,\pi + i\,\omega) (70\,560\,000\,\pi^2 + 8400\,i\,\pi\,\omega - \omega^2))$

In[228]:= `diff[ω] = frgangneu[ω] - frgang[ω]`

Out[228]= 0

Hier zum Abschluss nochmal mit **Mathematica** die ganze Faltung $X = G * F$ als Vektorfunktion in Abhängigkeit vom Zeitparameter t und nochmals der Frequenzgang als Fouriertransformierte der Komponente $X[[3]]$, welche die Impulsantwort des Filters ist.

In[229]:= `X = Table[Sum[ComplexExpand[Convolve[G[[k, m]], F[[m]], t, s]], {m, 1, 3}], {k, 1, 3}] /. s \rightarrow t`

Out[229]= $\{592\,704\,000\,000\,e^{-8400\,\pi\,t}\,\pi^3\,\text{UnitStep}[t] + 395\,136\,000\,000\,\sqrt{3}\,e^{-4200\,\pi\,t}\,\pi^3\,\text{Sin}[4200\,\sqrt{3}\,\pi\,t]\,\text{UnitStep}[t], 70\,560\,000\,e^{-8400\,\pi\,t}\,\pi^2\,\text{UnitStep}[t] - 70\,560\,000\,e^{-4200\,\pi\,t}\,\pi^2\,\text{Cos}[4200\,\sqrt{3}\,\pi\,t]\,\text{UnitStep}[t] + 70\,560\,000\,\sqrt{3}\,e^{-4200\,\pi\,t}\,\pi^2\,\text{Sin}[4200\,\sqrt{3}\,\pi\,t]\,\text{UnitStep}[t], 8400\,e^{-8400\,\pi\,t}\,\pi\,\text{UnitStep}[t] - 8400\,e^{-4200\,\pi\,t}\,\pi\,\text{Cos}[4200\,\sqrt{3}\,\pi\,t]\,\text{UnitStep}[t] + 2800\,\sqrt{3}\,e^{-4200\,\pi\,t}\,\pi\,\text{Sin}[4200\,\sqrt{3}\,\pi\,t]\,\text{UnitStep}[t]\}$

In[230]:= `X[[3]]`

Out[230]= $8400\,e^{-8400\,\pi\,t}\,\pi\,\text{UnitStep}[t] - 8400\,e^{-4200\,\pi\,t}\,\pi\,\text{Cos}[4200\,\sqrt{3}\,\pi\,t]\,\text{UnitStep}[t] + 2800\,\sqrt{3}\,e^{-4200\,\pi\,t}\,\pi\,\text{Sin}[4200\,\sqrt{3}\,\pi\,t]\,\text{UnitStep}[t]$

In[231]:= `FourierTransform[X[[3]], t, ω , FourierParameters \rightarrow {1, -1}]`

(* Frequenzgang wie in Beispiel 1 in Abschnitt 1.1 oben *)

Out[231]= $(592\,704\,000\,000\,\pi^3) / ((8400\,\pi + i\,\omega) (70\,560\,000\,\pi^2 + 8400\,i\,\pi\,\omega - \omega^2))$

Interessierte Leser mögen nun selbst aktiv werden und versuchen, mit den besprochenen Methoden eigenständig einige Aufgaben zu lösen.

Das Grundlösungsverfahren behält auch für partielle lineare Differentialgleichungen mit konstanten Koeffizienten seine Gültigkeit. Insbesondere gilt der berühmte **Satz von Malgrange-Ehrenpreis**, dass jede solche DGL eine Fundamentallösung besitzt, d.h. lösbar ist für alle rechten Seiten, die man mit einer solchen Grundlösung falten kann, also etwa für alle mit kompaktem Träger.

Einen elementaren Beweis dieses Schlüsselresultats können Sie in [1] finden.

Auf Beispiele zu partiellen Differentialgleichungen mit *Mathematica* gehe ich in einem späteren Notebook ein.

Literaturhinweis : Bei mehr Interesse am Thema mit Anwendungen der Distributionentheorie auf lineare Systeme empfehle ich neben [1]

[2] A.H. Zemanian Distribution Theory And Transform Analysis, Dover 1965